– 333 (revised) –

Stochastic calculus in *Mathematica*: software and examples

by

W.S. Kendall

May 1998, revised January 2003

DEPARTMENT OF STATISTICS

UNIVERSITY OF WARWICK

Stochastic calculus in *Mathematica*: software and examples

W.S. Kendall*

January 28, 2003

Abstract: This research report describes *Mathematica* notebooks available on the web at http://www.warwick.ac.uk/statsdept/Staff/WSK/. They range from a reference notebook which displays the source code of *Itovsn3*, through a simple introductory notebook, to notebooks which describe examples relating to problems in mathematical finance, coupling theory, and statistical shape.

Keywords: COMPUTER ALGEBRA; ITÔ CALCULUS; SEMIMARTINGALE; STOCHASTIC ITÔ CALCULUS.
AMS Subject Classification (2000): 60H05

Introduction

The computer package *Itovsn3* was initially developed for the computer algebra package *REDUCE* (Kendall 1987, 1988, 1991c), and then re-implemented in *Mathematica* Kendall (1993b) and then in *AXIOM* Kendall (2001) using the information-hiding features of these packages. In this research report we present seven *Mathematica* notebooks which use *Itovsn3*. The intention is to provide material for people who want to understand *Itovsn3* well enough to use it for their own purposes.

The notebooks in question are as follows.

- `Intro.nb`: this introduces the *Itovsn3* constructs and gives a simple example of programming to compute Itô integrals;
- `Reference.nb`: a non-executable notebook displaying the main contents of the Itovsn3 package in a convenient reference form;
- `Bessel.nb`: calculations with the three-dimensional Bessel process, up to computation of hitting probabilities and heat kernel;

*Research supported by EPSRC research grants GR/71677, GR/L56831, GR/M75785 and BBSRC grant 88/E08527

- `Itoarea.nb`: calculation of the moment generating function of Lévy stochastic area following an idea due to Svante Jansson;
- `BlackScholes.nb`: using *Itovsn3* to derive the celebrated Black-Scholes formula from mathematical finance;
- `MardiaDryden.nb`: derivation of the Mardia-Dryden distribution from statistical shape theory;
- `Reflect.nb`: a notebook showing the original computer algebra argument for an elegant and helpful representation for reflection-coupled Brownian motions in a half-plane.

As an appendix there is a listing of the current *Itovsn3* package.

The format of the report is as follows: separate sections contain hyperlinks to each *Mathematica* notebooks, each with a preliminary paragraph describing the basic features of the notebook and supplying extra references. (If you have access to *Mathematica* then you may choose to configure your browser to launch the notebooks on demand; but note you will need also to download `Itovsn3.m` and place it on your path!) The report concludes with a brief summary.

The terminology "*Itovsn3*" arises by combining "Itô" (in honour of Kiyosi Itô) and a contraction of "version 3", at which version level package development stabilized . . . .

1 `Intro.nb`: an introductory notebook

The first part of this notebook is concerned with various ways of introducing semimartingales, ranging from the most general through to specific procedures for introduction of standard Brownian motions. Notice in particular the versatile and useful `Itosde` procedure, which corresponds exactly to definition of a semimartingale using a stochastic differential equation. In the example `Itosde` is used to define Z as the solution to

$$\mathrm{d}_{\mathbb{I}} Z \quad = \quad Y \,\mathrm{d}_{\mathbb{I}} X - X \,\mathrm{d}_{\mathbb{I}} Y \tag{1.1}$$

so that Z is (proportional to) the Itô stochastic area corresponding to the two-dimensional Brownian motion (X, Y).

In the second part of the notebook there is a demonstration of simple programming using *Itovsn3*. An algorithm is implemented which rewrites Itô integrals involving polynomials of time t and a simple Brownian motion B, rewriting them in terms of classical integrals using only $\mathrm{d}t$ and not $\mathrm{d}_{\mathbb{I}} B$ (though of course the integrands still may contain instances of B!). A partial check of the algorithm is conducted using the famous relationship of Itô integrals to Hermite polynomials.

Link to Intro.nb

Note that `Itovsn3.m` is needed for this and the other notebooks.

Link to Itovsn3.m

2 `Reference.nb`: reference to *Itovsn3* features

This notebook is completely inactive, but uses the "folding" feature of *Mathematica* notebooks (the hidden cell feature) to produce a convenient on-line reference to the construction of *Itovsn3*. A list is given of all procedures of *Itovsn3*, grouped thematically in sections. Associated hidden calls contain indications of useage, brief textual notes, and listings of some of the relevant source code.

Link to Reference.nb

3 `Bessel.nb`

The Bessel process is the simplest non-trivial variation of the theme of standard Brownian motion. The Bessel process R of k degrees of freedom, often written BESS(k), is most readily defined for positive integers k by a *distance representation*: R is distributed as the Euclidean distance from the origin of a k-dimensional Brownian motion (each of the k coordinates is an independent standard Brownian motion). A special case ($k = 3$) is considered in this notebook.

From this representation one can use stochastic calculus to deduce the general stochastic differential equation

$$\mathrm{d}_{\mathbb{I}} R \quad = \quad \mathrm{d}_{\mathbb{I}} W + \frac{k-1}{2R}\,\mathrm{d}t \tag{3.1}$$

which can be used to define the Bessel process for positive real values of the "dimension" k.

In the notebook we compute hitting probabilities and transition kernel for a BESS(3) process. Much of this generalizes to other dimensions (though closed forms for the transition kernel can be expected only for odd integral k). However automatic derivation of Equation (3.1) for general integer k is in fact a complicated matter for computer algebra packages; it requires computation using "symbolic dimension" k. Eventually this leads to interesting, complicated, and not completely resolved issues concerning automatic simplification. See Kendall (1990) for a way of carrying out this sort of computation in *Itovsn3*, and Kendall (1992) for similar but more involved computations in statistical asymptotics, and a discussion of the simplification issues involved.

More advanced investigation of Bessel processes using *Itovsn3* may be found in Kendall (1991b).

Link to Bessel.nb

4 `Itoarea.nb`: calculation of the moment generating function of the Itô stochastic area

Paul Lévy derived a beautiful formula for the moment generating function of the Itô stochastic area defined in Equation (1.1). Svante Jansson developed a method of proving this using stochastic calculus, as described in Protter (1990, Theorem 42), and a treatment using *REDUCE Itovsn3* is to be found in Kendall (1993a). This notebook translates the *REDUCE* version to *Mathematica* and exploits *Mathematica*'s `DSolve` facility for solving the relevant differential equations.

Link to Itoarea.nb

5 `BlackScholes.nb`: pricing options

Financial mathematics is a natural application area for *Itovsn3*, not as yet exploited. The original *Mathematica* implementation of *Itovsn3* was initially applied in Kendall (1993b) to provide a computer algebra treatment of a hedging problem originally solved by Duffie and Richardson, and in the same volume a diffusion-based implementation of stochastic calculus, Steele and Stine (1993), was used to derived the celebrated Black-Scholes option pricing formula. Here we give an *Itovsn3* treatment of Black-Scholes. A derivation using the *AXIOM* implementation is to be found in Kendall (1998a).

Link to BlackScholes.nb

6 `MardiaDryden.nb`: statistics of shape

The initial application area of the original *REDUCE*-based implementation of *Itovsn3* was to the *statistical theory of shape* as expounded in Stoyan, Kendall, and Mecke (1995), Small (1996), Dryden and Mardia (1998), Barden, Carne, Kendall, and Le (1999). The original work (Kendall 1987, 1988) is still being developed, for example in Kendall (1998b). Here we use stochastic differential equation theory to derive the *Mardia-Dryden distribution* from shape theory (Mardia and Dryden (1989)). Note that there are other stochastic calculus approaches to this, for example Kendall (1991a), Le (1991). The attraction of the approach described here is that it adopts a rather empirical search method based loosely on heat-kernel asymptotics. A similar method is used to derive approximations to a shape distribution in Kendall (1998b).

An implementation of automatic diffusion geometry within the *REDUCE* version of *Itovsn3* has been used to give very rapid derivations of some shape geometries: see Kendall (1991c).

Link to MardiaDryden.nb

7 `Reflect.nb`: a coupling example

The final example is a genuine research contribution to the theory of coupling of random processes. In the course of the investigation reported in (Burdzy and Kendall 2000), motivated by problems in perfect simulation (Propp and Wilson (1996), Kendall (1998c)), it was required to gain insight into the behaviour of two reflection-coupled Brownian motions evolving within a half-plane with reflecting boundary conditions. An elegant and simple representation was derived using the *REDUCE* implementation of *Itovsn3* as an exploratory tool. In fact the representation is obvious when considered with the advantage of hindsight; nevertheless its discovery is directly attributable to *Itovsn3*.

A more extensive and substantial approach to a coupling problem, using a combination of algebraic, graphical, and numerical techniques all controlled by *REDUCE Itovsn3*, is reported in Arous, Cranston, and Kendall (1995).

Link to Reflect.nb

8 Conclusion

It is interesting to note the common features of these notebooks. Many of them (`Bessel.nb`, `BlackScholes.nb`, `Itoarea.nb`, `MardiaDryden.nb`,) can be viewed as using semimartingale stochastic calculus, particularly local martingale theory, to find solutions to stochastic differential equations. However this does not exhaust the possibilities of *Itovsn3*; consider that in other cases (`Intro.nb`, `Reflect.nb`) it is used to construct algorithms for solving stochastic calculus problems, or to find special and informative representations of geometrically-defined semimartingales. The power and attraction of *Itovsn3* lies in the way in which it relates the computer algebra “front-end” (the way in which expressions are input into the package) very closely to a familiar, expressive, and powerful stochastic calculus formalism. The *AXIOM* implementation of (Kendall 1998a, 2001) carries this further still, by using *AXIOM*’s facilities for programming mathematical structure so as to implement the actual algebra of stochastic differentials: it would be a very useful programming exercise now to introduce these ideas into *Itovsn3* itself — though after such an upgrade it had better be re-named *Itovsn**4***!

References[1]

Arous, G. B., M. Cranston, and W. Kendall [1995]. Coupling constructions for hypoelliptic diffusions: Two examples. In M. Cranston and M. Pinsky (Eds.), *Stochastic Analysis: Summer Research Institute July 11-30, 1993.*, Volume 57, Providence, RI Providence, pp. 193–212. American Mathematical Society.

Barden, D., T. Carne, D. Kendall, and H. Le [1999]. *The Statistical Theory of Shape and Its Applications*. Chichester: John Wiley & Sons.

Burdzy, K. and W. Kendall [2000, May]. Efficient Markovian couplings: examples and counterexamples. *The Annals of Applied Probability 10*(2), 362–409. Also: University of Warwick Department of Statistics Research Report 331.

Dryden, I. L. and K. Mardia [1998]. *The Statistical Analysis of Shape*. Chichester: John Wiley & Sons.

Kendall, D. [1991a]. The Mardia-Dryden shape distribution for triangles – a stochastic calculus approach. *Journal of Applied Probability 28*, 225–230.

Kendall, W. [1987]. Discussion of read paper by Clifford, Green and Pilling. *Journal of the Royal Statistical Society (Series B: Methodological) 49*, 286–287.

Kendall, W. [1988]. Symbolic computation and the diffusion of shapes of triads. *Advances in Applied Probability 20*, 775–797.

Kendall, W. [1990]. The Euclidean diffusion of shape. In D. Welsh and G. Grimmett (Eds.), *Disorder in Physical Systems*, Oxford, pp. 203–217. The Clarendon Press Oxford University Press. Also: University of Warwick Department of Statistics Research Report 161.

Kendall, W. [1991b]. Symbolic Itô calculus: An introduction. Research Report 217, Department of Statistics, University of Warwick.

Kendall, W. [1991c]. Symbolic Itô calculus: An overview. In N. Bouleau and D. Talay (Eds.), *Probabilités Numeriques*, Rocquencourt, pp. 186–192. INRIA.

Kendall, W. [1992]. Computer algebra and yoke geometry I: When is an expression a tensor? Research report, Department of Statistics, Warwick University.

[1] This is a rich hypertext bibliography. Journals are linked to their homepages. Stable URL links (as provided for example by JSTOR) have been added to entries where available. Access to such URLs may not be universal: in case of difficulty you should check whether you are registered (directly or indirectly) with the relevant provider.

Kendall, W. [1993a]. Computer algebra in probability and statistics. *statistica neerlandica 47*, 9–25.

Kendall, W. [1993b]. Doing stochastic calculus with *Mathematica*. In H. Varian (Ed.), *Economic and Financial Modeling with* Mathematica, pp. 214–238. New York: Springer-Verlag.

Kendall, W. [1998a]. *Itovsn3 in AXIOM: modules, algebras and stochastic differentials*. University of Warwick: Department of Statistics. Research Report 328.,

Kendall, W. [1998b]. A diffusion model for Bookstein triangle shape. *Advances in Applied Probability 30*(2), 317–334. Also: University of Warwick Department of Statistics Research Report 296.

Kendall, W. [1998c]. Perfect simulation for the area-interaction point process. In L. Accardi and C. Heyde (Eds.), *Probability Towards 2000*, New York, pp. 218–234. Springer-Verlag. Also: University of Warwick Department of Statistics Research Report 292.

Kendall, W. [2001]. Symbolic Itô calculus: an ongoing story. *Statistics and Computing 11*, 25–35. Also: University of Warwick Department of Statistics Research Report 327.

Le, H. [1991]. A stochastic calculus approach to the shape distribution induced by a complex normal model. *Math. Proc. Camb. Phil. Soc. 109*, 221–228.

Mardia, K. and I. Dryden [1989]. Shape distributions for landmark data. *Advances in Applied Probability 21*, 742–755.

Propp, J. and D. Wilson [1996]. Exact sampling with coupled Markov chains and applications to statistical mechanics. *Random Structures and Algorithms 9*, 223–252.

Protter, P. [1990]. *Stochastic Integration and Differential Equations: A New Approach*. New York - Berlin: Springer-Verlag.

Small, C. [1996]. *The statistical theory of shape*. New York-Heidelberg-Berlin: Springer-Verlag.

Steele, J. and R. Stine [1993]. *Mathematica* and diffusions. In H. Varian (Ed.), *Economic and Financial Modeling with* Mathematica, pp. 192–213. New York: Springer-Verlag.

Stoyan, D., W. Kendall, and J. Mecke [1995]. *Stochastic geometry and its applications* (Second ed.). Chichester: John Wiley & Sons. (First edition in 1987 joint with Akademie Verlag, Berlin).

A The package Itovsn3.m

```
(*
    Itovsn3.m: a Mathematica package for
    Symbolic Ito calculus Itovsn3. Version 3.70,
      Copyright March, June 1992, April 1998, November 2002, January 2003
    Author:   Wilfrid S.Kendall.

    Version 3.68 was tested on
    Mathematica version 1, Macintosh, Windows 3.1;
    Mathematica version 2, Windows 3.1;
    Mathematica version 2.2.3, Windows 97;
    Mathematica version 3.0, Linux.

    Version 3.39 has been tested on
    Mathematica version 4.0.0.0, Windows 2000

The author disclaims any responsibility for problems experienced by
users as a result of using this software, but welcomes your feedback!
The software may not be copied or distributed for commercial gain
without written permission from the author, but may be passed to
others if left intact in its present form.

Comments please to  w.s.kendall@warwick.ac.uk  or to

Prof W S Kendall,
Department of Statistics,
University of Warwick,
Coventry CV4 7AL, UK.
*)

BeginPackage["Itovsn3`"]

AddDrift::usage =
"AddDrift[dX,DriftdX]  sets  Drftbydt[dX] = DriftdX/dt,  so that\n
(for example)  Drift[dX]  yields  DriftdX."

AddFixed::usage =
"AddFixed[t0,X,X0]  sets  Fixed[t0,X] = X0,  so that\n
(for example)  InitialValue[t0,X]  yields  X0."

AddQuadVar::usage =
"AddQuadVar[dX,dY,xdt]  sets  Brktbydt[dX,dY] = xdt/dt,  altering \n
the effect of substitution using  ItoMultiplications,  so that\n
(for example)  ItoExpand[dX dY]  yields  xdt.  Similarly for\n
AddQuadVar[dX dY,xdt]  and  AddQuadVar[dX^2,xdt]."

Brktbydt::usage =
"Brktbydt[dX,dY]  is a placeholder for the FORMAL quotient by  dt  of\n
the bracket differential  dX dY."

BrownBasis::usage =
"BrownBasis[SemimartingaleList,InitialValueList]  introduces and sets\n
up second- and first-order structure for an independent set of Brownian\n
basic semimartingales, identifiers in  SemimartingaleList,  with initial \n
value expressions given by  InitialValueList.  It uses  BrownSingle[X,X0] \n
as a supplementary procedure. Corresponding basic stochastic differential\n
identifiers are created by prepending 'd' to the  SemimartingaleList  \n
identifiers."

BrownSingle::usage =
"BrownSingle[X,X0]  introduces and sets up a single Brownian basic\n
semimartingale identifier  X  (creating the semimartingale differential\n
identifier  dX)  with initial value expression  X0."

Drftbydt::usage =
"Drftbydt[dX]  is a placeholder for the FORMAL quotient by  dt
of Drift of  dX.\n"
```

```
Drift::usage =
"Drift[sd]  computes the drift differential of the stochastic\n
differential expression  sd.  NOTE that  Drift  assumes that  sd  is\n
genuinely a stochastic differential expression."

Fixed::usage =
"Fixed[t0,X]  is a placeholder for the fixed value of the basic \n
semimartingale X  at time  t0  (usually  t0=0)."

InitialValue::usage =
"InitialValue[t0,f]  computes the value of the expression  f  \n
at time  t0."

Introduce::usage =
"Introduce[Smgl,dSmgl]  introduces the basic semimartingale identifier \n
Smgl  with associated basic stochastic differential identifer  dSmgl.  \n
Attempts to reintroduce semimartingale or stochastic differential\n
identifiers are reported as errors."

ItoD::usage =
"ItoD[f]  computes the stochastic differential of the semimartingale\n
expression  f.  NOTE that  ItoD  assumes that  f  is a genuine\n
semimartingale expression!"

ItoExpand::usage =
"ItoExpand[sd]  computes simplication of the stochastic differential\n
products in the expression  sd  which is its argument."

ItoInit::usage =
"ItoInit[t,dt]  starts things off with basic structures, using the \n
identifier  t  for time variable and the identifier  dt  for its \n
differential."

ItoIntegral::usage =
"ItoIntegral[sd]  represents the Ito integral of the stochastic\n
differential expression  sd."

ItoReset::usage =
"ItoReset[t,dt]  resets all structures, using  ItoInit[t,dt]."

Itosde::usage =
"Itosde[X,dX==sd,X0]  introduces and sets up a basic semimartingale \n
identifier X  with basic stochastic differential identifier  dX  and \n
initial value expression  X0,  and satisfying the second- and first- \n
order structure implied by the stochastic differential equation  dX==sd."

ItoStatus::usage =
"ItoStatus[]  reports current structures."

RandomQ::usage =
"RandomQ[x,sdl]==True  if  x  is an expression in semimartingales or\n
stochastic differentials excluding those given in  sdl."

BSDQ::usage =
"BSDQ[x,sdl]==True  if  x  is a basic stochastic differential\n
excluding those given in  sdl."

(* ============================  *)

Begin["`private`"]

(* =>Implementing the Ito formula: *)
ItoD[f_] := Block[
{ ff =(f/.ItoIntegral[sdx_]->ItoIntegral[sdx,t]) },
ItoExpand[
                 (Dt[ff,t] dt + (1/2) Dt[ff,{t,2}] dt^2)/.Freeze[sd_]->sd ]
];
```

```
ItoExpand[sd_] := ((Expand[sd]
/. ItoIntegral[sdx_,t]->ItoIntegral[sdx])
/. ItoMultiplications);

(* =>Implementing Drift: *)
Drift[sd_] := Apply[Plus,Map[Coefficient[Expand[sd],#] Drftbydt[#] dt &,CSD]];

(* =>Updating first- and second-order structure: *)
AddQuadVar[dX_ dY_,xdt_] := AddQuadVar[dX,dY,xdt];
AddQuadVar[dX_^2,xdt_]   := AddQuadVar[dX,dX,xdt];
AddQuadVar[dX_,dY_,xdt_] := (Brktbydt[dX,dY] = xdt/dt);
AddDrift[dX_,DriftdX_]   := (Drftbydt[dX] = DriftdX/dt);

(* =>Finding initial value: *)
AddFixed[t0_,y_,y0_] := (Fixed[t0,y]=y0);
Derivative[1,0][Fixed][t0_,x_] := 0;
Derivative[0,1][Fixed][t0_,x_] := 0;
InitialValue[t0_,x_] :=
((((x /. Map[#->MayFix[t0,#]&,CS])
/. ItoIntegral[y_] ->MayFix[t0,ItoIntegral[y]])
/. MayFix[a_,y_] -> MayFix[a,(y/.MayFix[t0,u_]->u)])
/. MayFix[t0,u_] -> Fixed[t0,u]);

(* =>Simplest properties of ItoIntegral: *)
Derivative[0,2][ItoIntegral][sd_,t_] := 0;
Derivative[0,1][ItoIntegral][sd_,t_] := Freeze[sd]/ItoD[t];
Derivative[1][Freeze][sd_] := 0;
Derivative[1,0][ItoIntegral][sd_,t_] := 0;

(* =>Introducing basic semimartingale
and associated stochastic differential: *)
Introduce[X_, dX_] :=
If[Not[MemberQ[CS,X]] && Not[MemberQ[CSD,dX]],
(
X/:  Dt[X,t] = dX/dt;
dX/: Dt[dX,t] = 0 ;
        ItoIntegral[dX] = X - Fixed[0,X];
CSD = Prepend[CSD,dX];
CS = Prepend[CS,X];
ItoMultiplications = Join[
Map[(dX # -> Brktbydt[dX,#] dt)&,CSD],
ItoMultiplications ];
AddQuadVar[dX dt,0];
CSD
), "Attempt to re-introduce semimartingale or stochastic differential!"
];

(* =>Initialization: *)
ItoInit[time_,dtime_] :=
(
t = time;
dt = dtime;
SetAttributes[Brktbydt,Orderless];
CS = {};
CSD = {};
ItoMultiplications = {};
Introduce[t,dt];
AddDrift[dt,dt];
ItoIntegral[0] = 0;
Fixed[t0_,t] = t0;
Fixed[0,ItoIntegral[y_]] = 0;
TableForm[{
"Itovsn3  initialized",
SequenceForm["with time semimartingale ",t],
SequenceForm["and time differential ",dt]
}]
);
```

```
(* =>Resetting stochastic calculus structures: *)
ItoReset[time_,dtime_] :=
(
Map[ItoClear[#,ItoD[#]]&,CS];
Clear[Fixed];
Clear[Brktbydt];
Clear[Drftbydt];
        Clear[ItoIntegral];
TableForm[Prepend[First[ItoInit[time,dtime]],"Itovsn3  resetting ..."]]
);

(* =>Test if x involves semimartingales or differentials except
        those given in the second list (presumed deterministic!)*)
RandomQ[x_,sdl_] :=
        Not[FreeQ[x,Apply[Alternatives,Complement[Union[CS,CSD],sdl]]]]

(* =>Test if x is a basic stochastic differential excepting
        those given in the second list*)
BSDQ[x_,sdl_] :=
        MemberQ[Complement[CSD,sdl],x]

(* =>Clear aspects of semimartingale  X  and differential  dX: *)
ItoClear[X_,dX_] :=
( dX/: Dt[dX,t] = .; X/:  Dt[X,t] = .; );

(* =>Reporting status: *)
ItoStatus[] :=
Print[ColumnForm[{
        "---------------",
"Summary of current structure of stochastic differentials",
"- - - - - - - -",
"Current second-order structure of semimartingale differentials:",
TableForm[Outer[ItoExpand[#1 #2]&,CSD,CSD],TableHeadings->{CSD,CSD}],
"- - - - - - - -",
"Current first-order structure of semimartingale differentials:",
TableForm[Map[Drift[#]&,CSD],
  TableDirections->Row,TableHeadings->{CSD,{"Drifts:"}}],
"- - - - - - - -",
"Current initial values:",
TableForm[Map[Fixed[0,#]&,CS],
TableDirections->Row,TableHeadings->{CS,{"Initially:"}}],
"---------------"
}]];

(* =>Brownian Basis: *)
BrownBasis[SemimartingaleList_,InitialValueList_] :=
(
Map[Apply[BrownSingle,#]&,
Transpose[{SemimartingaleList,InitialValueList}]];
BrownPairs[Map[ItoD,SemimartingaleList]]
);
BrownSingle[X_,X0_] :=
Block[
{dX=ToExpression[StringJoin["d",ToString[X]]]},
Introduce[X,dX];
AddQuadVar[dX^2,dt];
AddDrift[dX,0];
Fixed[0,X]=X0
];
BrownPairs[SL_] :=
If[Length[SL]>1,
Map[AddQuadVar[First[SL] #,0]&,Rest[SL]];BrownPairs[Rest[SL]]];

(* =>   Definition using Ito stochastic differential equations: *)
Itosde[X_,dX_==sd_,X0_] :=
(
Introduce[X,dX];
AddDrift[dX,Drift[sd]];
```

```
Fixed[0,X]=X0;
AddQuadVar[dX^2,ItoExpand[sd^2]];
Map[AddQuadVar[dX #,ItoExpand[sd #]]&,CSD];
);

End[]
(* ============================  *)

EndPackage[]
(* ============================  *)
```

Other University of Warwick Department of Statistics Research Reports authored or co–authored by W.S. Kendall.

161: The Euclidean diffusion of shape.
162: Probability, convexity, and harmonic maps with small image I: Uniqueness and fine existence.
172: A spatial Markov property for nearest–neighbour Markov point processes.
181: Convexity and the hemisphere.
202: A remark on the proof of Itô's formula for C^2 functions of continuous semimartingales.
203: Computer algebra and stochastic calculus.
212: Convex geometry and nonconfluent Γ-martingales I: Tightness and strict convexity.
213: The Propeller: a counterexample to a conjectured criterion for the existence of certain convex functions.
214: Convex Geometry and nonconfluent Γ-martingales II: Well–posedness and Γ-martingale convergence.
216: *(with E. Hsu)* Limiting angle of Brownian motion in certain two–dimensional Cartan–Hadamard manifolds.
217: Symbolic Itô calculus: an introduction.
218: *(with H. Huang)* Correction note to "Martingales on manifolds and harmonic maps."
222: *(with O.E. Barndorff-Nielsen and P.E. Jupp)* Stochastic calculus, statistical asymptotics, Taylor strings and phyla.
223: Symbolic Itô calculus: an overview.
231: The radial part of a Γ-martingale and a non-implosion theorem.
236: Computer algebra in probability and statistics.
237: Computer algebra and yoke geometry I: When is an expression a tensor?
238: *Itovsn3*: doing stochastic calculus with *Mathematica*.
239: On the empty cells of Poisson histograms.
244: *(with M. Cranston and P. March)* The radial part of Brownian motion II: Its life and times on the cut locus.
247: Brownian motion and computer algebra (Text of talk to BAAS Science Festival '92, Southampton Wednesday 26 August 1992, with screenshots of illustrative animations).
257: Brownian motion and partial differential equations: from the heat equation to harmonic maps (Special invited lecture, 49th session of the ISI, Firenze).
260: Probability, convexity, and harmonic maps II: Smoothness via probabilistic gradient inequalities.
261: *(with G. Ben Arous and M. Cranston)* Coupling constructions for hypoelliptic diffusions: Two examples.
280: *(with M. Cranston and Yu. Kifer)* Gromov's hyperbolicity and Picard's little theorem for harmonic maps.
292: Perfect Simulation for the Area-Interaction Point Process.
293: *(with A.J. Baddeley and M.N.M. van Lieshout)* Quermass-interaction pro-

cesses.

295: On some weighted Boolean models.
296: A diffusion model for Bookstein triangle shape.
301: COMPUTER ALGEBRA: an encyclopaedia article.
308: Perfect Simulation for Spatial Point Processes.
319: Geometry, statistics, and shape.
321: From Stochastic Parallel Transport to Harmonic Maps.
323: *(with E. Thönnes)* Perfect Simulation in Stochastic Geometry.
325: *(with J.M. Corcuera)* Riemannian barycentres and geodesic convexity.
327: Symbolic Itô calculus in *AXIOM*: an ongoing story.
328: *Itovsn3* in *AXIOM*: modules, algebras and stochastic differentials.
331: *(with K. Burdzy)* Efficient Markovian couplings: examples and counterexamples.
333: Stochastic calculus in *Mathematica*: software and examples.
341: Stationary countable dense random sets.
347: *(with J. Møller)* Perfect Metropolis-Hastings simulation of locally stable point processes.
348: *(with J. Møller)* Perfect implementation of a Metropolis-Hastings simulation of Markov point processes
349: *(with Y. Cai)* Perfect simulation for correlated Poisson random variables conditioned to be positive.
350: *(with Y. Cai)* Perfect implementation of simulation for conditioned Boolean Model via correlated Poisson random variables.
353: *(with C.J. Price)* Zeros of Brownian Polynomials.
371: *(with G. Montana)* Small sets and Markov transition densities.
382: *(with A. Brix)* Simulation of cluster point processes without edge effects.
391: *(with E. Thönnes, A. Bhalerao, R.G. Wilson)* A Bayesian approach to inferring vascular tree structure from 2D imagery.
392: *(with A. Bhalerao, E. Thönnes, R.G. Wilson)* Inferring vascular tree structure from 2D and 3D imagery.
402: *(with R.G. Wilson)* Ising models and multiresolution quad-trees.

Also see the following related preprints

317: E. Thönnes: Perfect Simulation of some point processes for the impatient user.
334: M.N.M. van Lieshout and E. Thönnes: A Comparative Study on the Power of van Lieshout and Baddeley's J-function.
359: E. Thönnes: A Primer on Perfect Simulation.
366: J. Lund and E. Thönnes: Perfect Simulation for point processes given noisy observations.

If you want copies of any of these reports then please email your requests to the secretary using <statistics@warwick.ac.uk> (mail address: the Department of Statistics, University of Warwick, Coventry CV4 7AL, UK).